

**ma  
the  
ma  
tisch**

**cen  
trum**

---

DEPARTMENT OF NUMERICAL MATHEMATICS

JUNE

NUMAL, A LIBRARY OF NUMERICAL PROCEDURES IN ALGOL 60

VOLUME 2. ALGEBRAIC EVALUATIONS

*Nu 2*

---

**amsterdam**

**1974**



SECTION : 2.2.1.1

(OCTOBER 1975)

PAGE 1

AUTHOR: C.G. VAN DER LAAN.

INSTITUTE: REKENCENTRUM RIJKSUNIVERSITEIT GRONINGEN.

RECEIVED: 741114.

BRIEF DESCRIPTION:

THIS SECTION CONTAINS THE PROCEDURES POL, TAYPOL, NORDERPOL AND  
DERPOL.

POL EVALUATES A POLYNOMIAL;

DERPOL EVALUATES THE FIRST K DERIVATIVES OF A POLYNOMIAL;

NORDERPOL EVALUATES THE FIRST K NORMALIZED DERIVATIVES OF A  
POLYNOMIAL (I.E. J-TH DERIVATIVE/(J FACTORIAL), J=0,1,...,K<=DEGREE;

TAYPOL EVALUATES  $X^*J*(J-TH DERIVATIVE)/(J FACTORIAL)$ ,  
J=0,1,...,K<=DEGREE.

KEYWORDS:

POLYNOMIAL EVALUATION,  
(NORMALIZED) DERIVATIVES,  
DERIVATIVES OF A POLYNOMIAL.

SUBSECTION: POL.

CALLING SEQUENCE:

THE HEADING OF THE PROCEDURE READS:  
"REAL""PROCEDURE" POL(N,X,A);  
"VALUE"N,X;"INTEGER"N;"REAL"X;"ARRAY"A;

POL:=A[0]+A[1]\*X+...+A[N]\*X\*\*N;

THE MEANING OF THE FORMAL PARAMETERS IS:

N: <ARITHMETIC EXPRESSION>;  
ENTRY: THE DEGREE OF THE POLYNOMIAL;  
X: <ARITHMETIC EXPRESSION>;  
ENTRY: THE ARGUMENT OF THE POLYNOMIAL;  
A: <ARRAY IDENTIFIER>;  
"ARRAY"A[0:N];

ENTRY: THE COEFFICIENTS OF THE POLYNOMIAL  
A[0]+A[1]\*X+...+A[N]\*X\*\*N.

PROCEDURES USED: NONE.



SECTION : 2.2.1.1

(OCTOBER 1975)

PAGE 2

RUNNING TIME : PROPORTIONAL TO N  
(N MULTIPLICATIONS AND ADDITIONS).

LANGUAGE : ALGOL 60.

## METHOD AND PERFORMANCE :

THE METHOD USED FOR EVALUATION IS HORNER'S RULE (SYNTHETIC DIVISION). THE ERROR GROWTH IS GIVEN BY A LINEAR FUNCTION OF THE DEGREE OF THE POLYNOMIAL (SEE VAN DER LAAN, STOER(1972) P. 29 (EX. 11) OR WILKINSON(1963) P. 36,37).

SUBSECTION : DERPOL.

## CALLING SEQUENCE :

THE HEADING OF THE PROCEDURE READS :

```
"PROCEDURE" DERPOL(N,K,X,A);
"VALUE" N,K,X; "INTEGER" N,K; "REAL" X; "ARRAY" A;
```

THE MEANING OF THE FORMAL PARAMETERS IS :

```
N: <ARITHMETIC EXPRESSION>;
   ENTRY: THE DEGREE OF THE POLYNOMIAL;
K: <ARITHMETIC EXPRESSION>;
   ENTRY: THE FIRST K DERIVATIVES ARE TO BE CALCULATED;
X: <ARITHMETIC EXPRESSION>;
   ENTRY: THE ARGUMENT OF THE POLYNOMIAL;
A: <ARRAY IDENTIFIER>;
   "ARRAY" A[0:N];
   ENTRY: THE COEFFICIENTS OF THE POLYNOMIAL
         A[0]+A[1]*X+...+A[N]*X**N;
EXIT: THE J-TH DERIVATIVE IS DELIVERED IN A[J], J=0,1,...,K<=
      DEGREE; THE OTHER ELEMENTS ARE GENERALLY ALTERED.
```

## PROCEDURES USED :

NORJERPOL = CP31242

RUNNING TIME : THE NUMBER OF ADDITIONS IS  $(K+1)*(N-K/2)$  AND  
THE NUMBER OF MULTIPLICATIONS IS N, IN FIRST ORDER.

LANGUAGE : ALGOL 60.

METHOD AND PERFORMANCE : SEE TAYPOL (THIS SECTION).



SECTION : 2.2.1.1

(OCTOBER 1975)

PAGE 3

SUBSECTION: NORDERPOL.

CALLING SEQUENCE:

THE HEADING OF THE PROCEDURE READS:

```
"PROCEDURE" NORDERPOL(N,K,X,A);  
"VALUE" N,K,X;"INTEGER" N,K;"REAL" X;"ARRAY" A;
```

THE MEANING OF THE FORMAL PARAMETERS IS:

```
N: <ARITHMETIC EXPRESSION>;  
   ENTRY: THE DEGREE OF THE POLYNOMIAL;  
K: <ARITHMETIC EXPRESSION>;  
   THE FIRST K NORMALIZED DERIVATIVES ARE TO BE CALCULATED  
   (I.E. (J-TH DERIVATIVE)/(J FACTORIAL), J=0,1,...,K<=DEGREE).  
X: <ARITHMETIC EXPRESSION>;  
   ENTRY: THE ARGUMENT OF THE POLYNOMIAL;  
A: <ARRAY IDENTIFIER>;  
   "ARRAY" A[0:N];  
   ENTRY: THE COEFFICIENTS OF THE POLYNOMIAL  
         A[0]+A[1]*X+...+A[N]*X**N;  
   EXIT: THE J-TH NORMALIZED DERIVATIVE IS DELIVERED IN A[J]  
         J=0,1,...,K<=DEGREE; THE OTHER ELEMENTS ARE GENERALLY  
         ALTERED.
```

PROCEDURES USED: NONE.

REQUIRED CENTRAL MEMORY: AN AUXILIARY ARRAY OF ORDER  $N + 1$  IS DECLARED.

RUNNING TIME: THE NUMBER OF ADDITIONS IS  $(K+1)*(N-K/2)$  AND  
THE NUMBER OF MULTIPLICATIONS/DIVISIONS IS  $2 * N + K$ .

LANGUAGE: ALGOL 60.

METHOD AND PERFORMANCE: SEE TAYPOL (THIS SECTION).



SECTION : 2.2.1.1

(OCTOBER 1975)

PAGE 4

SUBSECTION: TAYPOL.

CALLING SEQUENCE:

THE HEADING OF THE PROCEDURE READS:

```

"PROCEDURE" TAYPOL(N,K,X,A);
"VALUE" N,K,X;"INTEGER" N,K;"REAL" X;"ARRAY" A;

```

THE MEANING OF THE FORMAL PARAMETERS IS:

```

N: <ARITHMETIC EXPRESSION>;
   ENTRY: THE DEGREE OF THE POLYNOMIAL;
K: <ARITHMETIC EXPRESSION>;
   ENTRY: THE FIRST K TERMS  $X^{*J} \cdot (J\text{-TH DERIVATIVE}) / (J \text{ FACTORIAL})$ ,
          $J=0,1,\dots,K \leq \text{DEGREE}$ , ARE TO BE CALCULATED;
X: <ARITHMETIC EXPRESSION>;
   ENTRY: THE ARGUMENT OF THE POLYNOMIAL;
A: <ARRAY IDENTIFIER>;
   "ARRAY" A[0:N];
   ENTRY: THE COEFFICIENTS OF THE POLYNOMIAL
          $A[0]+A[1]*X+\dots+A[N]*X^{*N}$ ;
   EXIT: THE  $J\text{-TH TERM}$   $X^{*J} \cdot (J\text{-TH DERIVATIVE}) / (J \text{ FACTORIAL})$ , IS
         DELIVERED IN A[J],  $J=0,1,\dots,K \leq \text{DEGREE}$ ; THE OTHER
         ELEMENTS ARE GENERALLY ALTERED.

```

PROCEDURES USED: NONE.

RUNNING TIME: THE NUMBER OF ADDITIONS IS  $(K+1) \cdot (N-K/2)$  AND  
 THE NUMBER OF MULTIPLICATIONS IS  $2 \cdot N$ .

LANGUAGE: ALGOL 60.

METHOD AND PERFORMANCE:

THE METHOD OF EVALUATION IS GIVEN BY TRAUB AND SHAW(1972,1974).  
 LET  $X^{*J} \cdot (J\text{-TH DERIVATIVE OF THE POLYNOMIAL}) / (J \text{ FACTORIAL}) =$   
 $(J \text{ OVER } J) \cdot A[J] \cdot X^{*J} + (J+1 \text{ OVER } J) \cdot A[J+1] \cdot X^{*(J+1)} + \dots + (N \text{ OVER } J) \cdot$   
 $A[N] \cdot X^{*N}$ , THEN THE  $J\text{-TH DERIVATIVE}$  (JP TO A FACTOR) CAN BE OBTAINED  
 FROM THE BINOMIAL COEFFICIENTS FOLLOWED BY EVALUATION OF THE ABOVE  
 INPRODUCT. THE SHAW AND TRAUB ALGORITHM PERFORMS THE BUILDING UP OF  
 THE BINOMIAL COEFFICIENTS IMPLICITLY.  
 WE HAVE NOT IMPLEMENTED THE MORE SOPHISTICATED ALGORITHM, BASED  
 ON DIVISORS OF  $N+1$ , BECAUSE OF THE MORE COMPLEX APPEARANCE  
 OF THE IMPLEMENTATION AND BECAUSE OF THE DIFFICULTY IN CHOSING  
 THE MOST EFFICIENT DIVISOR. OUR (RESTRICTED) IMPLEMENTATION OF THE  
 ONE-PARAMETER FAMILY OF ALGORITHMS PRESERVES THE LINEAR NUMBER OF  
 MULTIPLICATIONS ( $2 \cdot N$  (NORDERPOL, TAYPOL) AND  $3 \cdot N$  (DERPOL)).  
 THE ABSOLUTE ERROR IS OF ORDER  $\max((N \text{ OVER } N) \cdot A[N] \cdot X^{*(N-K)}, \dots,$   
 $(N \text{ OVER } K) \cdot A[K])$ , FOR THE  $K\text{-TH NORMALIZED DERIVATIVE}$  (SEE VAN DER  
 LAAN OR WOZNIAKOWSKI).



## REFERENCES:

- [1]. SHAW, M. AND J. TRAUB:  
ON THE NUMBER OF MULTIPLICATIONS FOR THE EVALUATION OF A  
POLYNOM AND SOME OF ITS DERIVATIVES (21 P.).  
DEP. COMPUTER SCIENCE, CARNEGIE-MELLON(1972); OR JOURN. ACM,  
1974, VOL. 21, NO. 1, P. 161-167.
- [2]. STOER, J. :  
EINFUEHRUNG IN DIE NUMERISCHE MATHEMATIK 1.  
SPRINGER, 1972.
- [3]. VAN DER LAAN C.G. :  
ORTHOGONAL POLYNOMIALS IN NUMERICAL ANALYSIS.  
1. ERROR ANALYSIS OF RECURRENCE RELATIONS IN FLOATING-POINT  
COMPUTATION, (TO APPEAR).
- [4]. WILKINSON, J.H. :  
ROUNDING ERRORS IN ALGEBRAIC PROCESSES.  
HSO, NOTES ON APPLIED SCIENCES NO. 32, 1963.
- [5]. WOZNIAKOWSKI, H. :  
ROUNDING ERROR ANALYSIS FOR THE EVALUATION OF A POLYNOMIAL AND  
SOME OF ITS DERIVATIVES.  
SIAM J. OF NUM. AN. VOL 11, NO. 4, P. 780-787.

## EXAMPLE OF USE:

AS A FORMAL TEST OF THE PROCEDURE DERPOL THE DERIVATIVES OF THE  
POLYNOMIAL  $3*X^3-2*X^2+X-1$  ARE CALCULATED.

```

"BEGIN""ARRAY"A[0:3];
  "PROCEDURE"DERPOL(N,K,X,A);"CODE"31243;
  A[3]:=3;A[2]:=-2;A[1]:=1;A[0]:=-1;
  DERPOL(3,3,1,A); OUTPUT(61,"(
  ""THE 0-TH UNTIL AND INCLUDING THE 3-TH DERIVATIVES :""),
  4(BZDB)""),A[0],A[1],A[2],A[3]);
"END" EXAMPLE OF USE;

```

THE 0-TH UNTIL AND INCLUDING THE 3-TH DERIVATIVES : 1 16 14 18

## SOURCE TEXT(S):

```

"CODE" 31040;
"REAL" "PROCEDURE" POL(N,X,A);
"VALUE" N,X;"INTEGER" N;"REAL" X;"ARRAY" A;
"BEGIN" "REAL" R;
  R:= 0;
  "FOR" N:= N "STEP" -1 "UNTIL" 0 "DO"
  R:=R*X + A[N];
  POL:= R
"END" POL;
"EOP"

```



```

"CODE" 31241;
"PROCEDURE" TAYPOL (N,K,X,A);
"VALUE" N,K,X;
"INTEGER" N,K;"REAL" X;"ARRAY" A;
"IF" X^= 0 "THEN"
"BEGIN" "INTEGER" I,J,NM1;
"REAL" XJ,AA,H;
XJ:=1;
"FOR" J:= 1 "STEP" 1 "UNTIL" N "DO"
"BEGIN" XJ:=XJ*X;A[J]:=A[J]*XJ "END";
AA:=A[N];NM1:=N-1;
"FOR" J:= 0 "STEP" 1 "UNTIL" K "DO"
"BEGIN" H:=AA;
"FOR" I:= NM1 "STEP" -1 "UNTIL" J "DO"
H:= A[ I]:=A[I]+H
"END"
"END" "ELSE"
"FOR" K:= K "STEP" -1 "UNTIL" 1 "DO" A[K]:=0;
"EOP"

```

```

"CODE" 31242;
"PROCEDURE" NORDERPOL (N,K,X,A);
"VALUE" N,K,X;
"INTEGER" N,K;"REAL" X;"ARRAY" A;
"IF" X^= 0 "THEN"
"BEGIN" "INTEGER" I,J,NM1;
"REAL" XJ,AA,H;
"ARRAY" XX[0:N];
XJ:=1;
"FOR" J:= 1 "STEP" 1 "UNTIL" N "DO"
"BEGIN" XJ:=XX[J]:=XJ*X;A[J]:=A[J]*XJ "END";
H:=AA:=A[N];NM1:=N-1;
"FOR" I:= NM1 "STEP" -1 "UNTIL" 0 "DO" H:= A[I]:=A[I]+H;
"FOR" J:= 1 "STEP" 1 "UNTIL" K "DO"
"BEGIN" H:=AA;
"FOR" I:= NM1 "STEP" -1 "UNTIL" J "DO"
H:= A[ I]:=A[I]+H;
A[J]:=H/XX[J]
"END"
"END" NORDERPOL ;
"EOP"

```

```

"CODE" 31243;
"PROCEDURE" DERPOL (N,K,X,A);
"VALUE" N,K,X;
"INTEGER" N,K;"REAL" X;"ARRAY" A;
"BEGIN" "INTEGER" J; "REAL" FAC;
"PROCEDURE" NORDERPOL (N,K,X,A); "CODE" 31242;
FAC:=1;
NORDERPOL (N,K,X,A);
"FOR" J:= 2 "STEP" 1 "UNTIL" K "DO"
"BEGIN" FAC:=FAC*J;A[J]:=A[J]*FAC "END"
"END" DERPOL ;
"EOP"

```





AUTHOR: C.G. VAN DER LAAN  
 CONTRIBUTORS: C.G. VAN DER LAAN, M. VOORINTHOLT  
 INSTITUTE: REKENCENTRUM RIJKSUNIVERSITEIT GRONINGEN  
 RECEIVED: 760201

BRIEF DESCRIPTION:

THIS SECTION CONTAINS THREE PROCEDURES FOR THE EVALUATION OF ORTHOGONAL POLYNOMIALS, GIVEN BY A SET OF RECURRENCE COEFFICIENTS:  
 ORTPOL: EVALUATES THE ORTHOGONAL POLYNOMIAL,  
 ALLORTPOL: EVALUATES ALL ORTHOGONAL POLYNOMIALS LOWER THAN A GIVEN DEGREE,  
 SERORTPOL: EVALUATES A FINITE SERIES EXPRESSED IN ORTHOGONAL POLYNOMIALS.

KEYWORDS:  
 ORTHOGONAL POLYNOMIAL,  
 SERIES OF ORTHOGONAL POLYNOMIALS,  
 LINEAR THREE TERM (IN)HOMOGENEOUS RECURRENCE RELATION.

DATA AND RESULTS:  
 ORTHOGONAL POLYNOMIALS F CAN BE CHARACTERIZED BY A RECURRENCE RELATION OF THE FOLLOWING FORM

$$A1[K] * F[K+1](X) = (A2[K] + X * A3[K]) * F[K](X) - A4[K] * F[K-1](X),$$

WHERE AI[K] ARE REAL NUMBERS. SEE FOR INSTANCE TABLE 22.7 IN ABRAMOWITZ AND STEGUN (1964) FOR THE CLASSICAL ORTHOGONAL POLYNOMIAL. BY AN ELEMENTARY TRANSFORMATION, THE COEFFICIENTS IN THE RECURRENCE RELATION ABOVE CAN BE MADE SUCH THAT A1[K] \* A3[K] = 1. IN OUR ALGORITHMS WE USE THIS SIMPLIFIED FORM OF THE RECURRENCE RELATION, WHICH IS NOW GIVEN BY

$$P[K+1](X) = (X - B[K]) * P[K](X) - C[K] * P[K-1](X),$$

P[0](X) = 1, P[1](X) = X - B[0]. IN THIS WAY, WE OBTAIN A NORMALIZATION OF THE ORTHOGONAL POLYNOMIAL SUCH THAT THE LEADING COEFFICIENT IN THE EXPLICIT REPRESENTATION OF P[K](X) EQUALS 1.

AS A CONSEQUENCE THE FOLLOWING RELATION HOLDS BETWEEN THE VALUES OF P[N](X) OBTAINED BY OUR PROCEDURES (E.G. ORTPOL) AND THE VALUES OF F[N](X) OBTAINED BY THE REPRESENTATION IN ABRAMOWITZ AND STEGUN (1964),

$$ORTPOL[N](X) = P[N](X) = F[N](X) * \prod_{K=0}^{N-1} (A1[K] / A3[K])$$



WHERE  $A_1(K)$ ,  $A_3(K)$ ,  $F(N)$  ARE DETERMINED BY TABLE 22.7 IN ABRAMOWITZ AND STEGUN (1964). WE WILL REMARK THAT OVERFLOW/UNDERFLOW MAY OCCUR FOR LOWER N-VALUES AS A CONSEQUENCE OF OUR NORMALIZATION. IN ORDER TO AVOID MISTAKES BY OBTAINING THE RECURRENCE COEFFICIENTS THE FOLLOWING TABLE GIVES THE RECURRENCE COEFFICIENTS FOR THE CLASSICAL ORTHOGONAL POLYNOMIALS (NOTE THAT THE FIRST AND SECOND POLYNOMIAL IS DEFINED BY THE NORMALIZATION AND  $B(0)$ ):

POLYNOMIAL KIND	RECURRENCE COEFFICIENTS	
	B(K)	C(K)
CHEBYSHEV (1-ST KIND)	0	$1/2, K=1$ $1/4, K>1$
CHEBYSHEV (2-ND KIND)	0	$1/4$
LEGENDRE	0	$K^2/(4K^2-1)$
JACOBI	$-(\alpha^2-\beta^2)/((2K+\alpha+\beta)(2K+\alpha+\beta+2))$	$4(1+\alpha)(1+\beta)/((\alpha+\beta+2)(\alpha+\beta+3)), K=1$ $4K(K+\alpha)(K+\beta)/((2K+\alpha+\beta)^2(2K+\alpha+\beta+1)), K>1$
LAGUERRE	$2K+\alpha+1$	$K(K+\alpha)$
HERMITE	0	$K/2$

IN GENERAL THE RECURRENCE COEFFICIENTS MAY BE OBTAINED BY USE OF THE PROCEDURE RECCOF. THE ABOVE TABLE IS OBTAINED BY ADAPTION OF ABRAMOWITZ AND STEGUN (1964), TABLE 22.7, P. 782 AS FOLLOWS: (NOTE THAT  $N \geq 1$ ; FOR  $N=0$  CONSULT 22.4)

$$B(I) := -A_2(I)/A_3(I) \quad , \quad I=0,1,\dots,N-1$$

$$C(I) := A_4(I)/(A_3(I)*A_3(I-1))*A_1(I-1) \quad , \quad I=1,2,\dots,N-1.$$



LANGUAGE: ALGOL60.

METHOD AND PERFORMANCE:

THE ORTHOGONAL POLYNOMIAL  $P_{k+1}$  IS DEFINED BY

$$P_{k+1}(X) = (X - B[k]) * P_k(X) - C[k] * P_{k-1}(X) \quad , \quad k=1,2,\dots,N-1$$

WHERE  $B[0:N-1]$ ,  $C[1:N-1]$  CONTAIN THE RECURRENCE COEFFICIENTS AND  
 $P_1(X) = X - B[0]$ ,  $P_0(X) = 1$ .  
 THEN (SEE STOER 1972, P. 119)

$$P_N(X) := (1,0) * \text{PROD}_{k=1}^{N-1} \left[ \begin{array}{c} X - B[k] \\ 1 \end{array} \right] - C[k] \left[ \begin{array}{c} X - B[0] \\ 1 \end{array} \right]$$

FOR  $N \geq 1$ .

FOR SERORTPOL WE USE THE ALGORITHM

$$\left[ \begin{array}{c} R[k] \\ S[k] \end{array} \right] (X) = \left[ \begin{array}{c} 0 \\ 1 \end{array} \right] - C[k] \left[ \begin{array}{c} R[k+1] \\ S[k+1] \end{array} \right] (X) + \left[ \begin{array}{c} A[k-1] \\ 0 \end{array} \right]$$

FOR  $k=N-1, N-2, \dots, 1$  WITH INITIAL VALUES  $R[N] = A[N-1]$ ,  $S[N] = A[N]$ .  
 THEN

$$\text{SERORTPOL} := (P_0(X), P_1(X)) * \left[ \begin{array}{c} R[1] \\ S[1] \end{array} \right] (X)$$

(SEE ALSO LUKE, 1969, P. 327).

THE MAGNITUDE OF THE ERROR IS A LINEAR FUNCTION OF THE DEGREE  $N$   
 (SEE VAN DER LAAN (TO APPEAR)).

## REFERENCES:

- [1] ABRAMOWITZ, M. & I. STEGUN (1964):  
 HANDBOOK OF MATHEMATICAL FUNCTIONS.  
 NATIONAL BUREAU OF STANDARDS, WASHINGTON D.C.
- [2] LAAN, C.G. VAN DER (TO APPEAR):  
 ORTHOGONAL POLYNOMIALS IN NUMERICAL ANALYSIS I.  
 ERROR ANALYSIS OF RECURRENCE RELATIONS.
- [3] LUKE, Y.L. (1969):  
 THE SPECIAL FUNCTIONS AND THEIR APPROXIMATIONS I.  
 ACADEMIC PRESS, LONDON, NEW YORK.
- [4] STOER, J. (1972):  
 EINFUEHRUNG IN DIE NUMERISCHE MATHEMATIK I.  
 SPRINGER VERLAG, BERLIN.



## SECTION : 2.2.2.1

(NOVEMBER 1976)

PAGE 4

## SUBSECTION: ORTPOL.

## CALLING SEQUENCE:

THE DECLARATION OF THE PROCEDURE IN THE CALLING PROGRAM READS:

```
"REAL" "PROCEDURE" ORTPOL(N,X,B,C); "VALUE" N,X;
"INTEGER" N; "REAL" X; "ARRAY" B,C;
"CODE" 31044;
```

```
ORTPOL: DELIVERS THE VALUE OF THE ORTHOGONAL POLYNOMIAL
P[N](X) OF DEGREE N FOR THE ARGUMENT X AS
DETERMINED BY THE RECURRENCE COEFFICIENTS B AND C.
```

THE MEANING OF THE FORMAL PARAMETERS IS:

```
N: <ARITHMETIC EXPRESSION>;
ENTRY: THE DEGREE OF THE POLYNOMIAL;
X: <ARITHMETIC EXPRESSION>;
ENTRY: THE ARGUMENT OF THE ORTHOGONAL POLYNOMIAL;
B,C: <ARRAY IDENTIFIER>;
"ARRAY" B[0:N-1], C[1:N-1];
ENTRY: THE RECURRENCE COEFFICIENTS (SEE DATA AND RESULTS).
```

## SUBSECTION: ALLORTPOL.

## CALLING SEQUENCE:

THE DECLARATION OF THE PROCEDURE IN THE CALLING PROGRAM READS:

```
"PROCEDURE" ALLORTPOL(N,X,B,C)RESULTS:(P);
"VALUE" N,X; "INTEGER" N; "REAL" X; "ARRAY" B,C,P;
"CODE" 31045;
```

THE MEANING OF THE FORMAL PARAMETERS IS:

```
N,X,B,C: SEE ORTPOL (THIS SECTION);
P: <ARRAY IDENTIFIER>;
"ARRAY" P[0:N];
EXIT: P[K] CONTAINS, FOR THE ARGUMENT X, THE VALUE OF
THE K-TH ORTHOGONAL POLYNOMIAL P[K](X) AS
DEFINED BY THE RECURRENCE COEFFICIENTS.
```



SECTION : 2.2.2.1

(NOVEMBER 1976)

PAGE 5

SUBSECTION: SERORTPOL.

CALLING SEQUENCE:

THE DECLARATION OF THE PROCEDURE IN THE CALLING PROGRAM READS:

```
"REAL" "PROCEDURE" SERORTPOL(N,X,B,C,A);
"VALUE" N,X; "INTEGER" N; "REAL" X; "ARRAY" B,C,A;
"CODE" 31047;
```

SERORTPOL: DELIVERS THE VALUE OF POLYNOMIAL  
 $A[0] + A[1]*P[1](X) + \dots + A[N]*P[N](X)$   
 WHERE  $P[K](X)$  IS THE K-TH ORTHOGONAL POLYNOMIAL.

THE MEANING OF THE FORMAL PARAMETERS IS:

N,X,B,C: SEE ORTPOL (THIS SECTION);

A: &lt;ARRAY IDENTIFIER&gt;;

"ARRAY" A[0:N];

ENTRY: THE COEFFICIENTS OF THE SERIES EXPANSION

 $A[0] + A[1]*P[1](X) + \dots + A[N]*P[N](X)$ WHERE  $P[K](X)$  IS THE K-TH ORTHOGONAL POLYNOMIAL  
AS DEFINED BY THE RECURRENCE COEFFICIENTS.

EXAMPLE OF USE:

AS AN EXAMPLE THE FOLLOWING PROGRAM DELIVERS THE VALUES OF THE  
 LAGUERRE POLYNOMIAL OF DEGREES 0,1,2,3,4,5 FOR  $X=0$  BY MEANS OF THE  
 PROCEDURE ALLORTPOL ( $B[K]=2*K+1$ ,  $C[K]=K**2$ ):

```
"BEGIN" "ARRAY" B[0:4], C[1:4], P[0:5];
"INTEGER" I;
"PROCEDURE" ALLORTPOL(N,X,B,C,P); "CODE" 31045;
B[0]:=1;
"FOR" I:=1 "STEP" 1 "UNTIL" 4 "DO"
"BEGIN" B[I]:=2*I+1;
C[I]:=I**2;
"END" I;
ALLORTPOL(5,0,B,C,P);
OUTPUT(61,"(")",(P[I],I:=0:5));
"END" PROGRAM;
```

RESULTS (NOTE DIFFERENCE WITH FIGURE 22.9 IN ABRAMOWITZ AND STEGUN  
 (1964) BECAUSE OF THE NORMALIZATION):

+1.0 -1.0 +2.0 -6.0 +24.0 -120.0

SOURCETEXTS:

```
"CODE" 31044;
"REAL" "PROCEDURE" ORTPOL(N,X,B,C);
"VALUE" N,X; "INTEGER" N; "REAL" X; "ARRAY" B,C;
"COMMENT" LET THE ORTHOGONAL POLYNOMIAL BE DEFINED BY
P[K+1](X) = (X-B[K])P[K](X) - C[K]P[K-1](X), K=1,2,...,N-1
```



WHERE  $B[0:N-1], C[1:N-1]$  ARE THE RECURRENCE COEFFICIENTS  
AND  $P[1](X) = X - B[0], P[0](X) = 1,$   
(SEE STOER 1972, PAR 3.5 P. 119),  
THEN

$$P[N](X) := (1,0) * \text{PROD}_{K=1}^{N-1} \left[ \frac{X-B[K] - C[K]}{1} \right] * \left[ \frac{X-B[0]}{1} \right]$$

THE RECURRENCE COEFFICIENTS MAY BE OBTAINED BY USE  
OF THE PROCEDURE RECCOF OR ADAPTING IN  
ABRAMOWITZ AND STEGUN, 1964 THE TABLE 22.7 P. 782 AS FOLLOWS  
(NOTE THAT IN THAT TABLE  $N \geq 1$ , FOR  $N=0$  CONSULT 22.4)  
 $B[I] := -A2(I)/A3(I) \quad , I=0,1,2,\dots,N-1$   
 $C[I] := A4(I)/(A3(I)*A3(I-1))*A1(I-1) \quad , I=1,2,\dots,N-1;$

```
"IF" N=0 "THEN" ORTPOL:=1 "ELSE"
"BEGIN" "INTEGER" K,L; "REAL" R,S,H;
  R:=X-B[0]; S:=1; L:=N-1;
  "FOR" K:=1 "STEP" 1 "UNTIL" L "DO"
"BEGIN" H:=R;
  R:=(X-B[K])*R-C[K]*S;
  S:=H;
"END";
  ORTPOL:=R;
"END" ORTPOL;
  "EQP"
"CODE" 31045;
"PROCEDURE" ALLORTPOL(N,X,B,C)RESULTS:(P);
"VALUE" N,X; "INTEGER" N; "REAL" X; "ARRAY" B,C,P;
"COMMENT" LET THE ORTHOGONAL POLYNOMIAL BE DEFINED BY
  P[K+1](X) = (X-B[K])P[K](X) - C[K]P[K-1](X), K=1,2,...,N-1
  WHERE B[0:N-1],C[1:N-1] ARE THE RECURRENCE COEFFICIENTS AND
  P[1](X) = X - B[0], P[0](X) = 1, THEN
```

$$P[I](X) := (1,0) * \text{PROD}_{K=1}^{N-1} \left[ \frac{X-B[K] - C[K]}{1} \right] * \left[ \frac{X-B[0]}{1} \right]$$

FOR  $I=2,3,\dots,N$ .  
THE RECURRENCE COEFFICIENTS MAY BE OBTAINED BY USE OF THE  
PROCEDURE RECCOF OR ADAPTING IN ABRAMOWITZ AND STEGUN, 1964  
THE TABLE 22.7 P. 782 AS FOLLOWS (NOTE THAT IN TABLE  $N \geq 1$ ,  
FOR  $N=0$  CONSULT 22.4)

$$B[I] := -A2(I)/A3(I) \quad , I=0,1,\dots,N-1$$

$$C[I] := A4(I)/(A3(I)*A3(I-1))*A1(I-1) \quad , I=1,2,\dots,N-1;$$

```
"IF" N=0 "THEN" P[0]:=1 "ELSE"
"BEGIN" "INTEGER" K,K1; "REAL" R,S,H;
  R:=P[1]:=X-B[0]; S:=P[0]:=1; K:=1;
  "FOR" K1:=2 "STEP" 1 "UNTIL" N "DO"
"BEGIN" H:=R; P[K1]:=R:=(X-B[K])*R-C[K]*S;
  S:=H; K:=K1;
"END";
"END" ALLORTPOL
  "EQP"
```



```

"CODE" 31047;
"REAL" "PROCEDURE" SERORTPOL(N,X,B,C,A);
"VALUE" N,X; "INTEGER" N; "REAL" X; "ARRAY" B,C,A;
"COMMENT" SERORTPOL:=A[0]P[0](X) + A[1]P[1](X) + ... + A[N]P[N](X),
  WHERE P[I](X), I=0,...,N ARE ORTHOGONAL POLYNOMIALS DEFINED BY
  THE RECURRENCE RELATION
    P[K+1](X) = (X-B[K])P[K](X) - C[K]P[K-1](X), K=1,2,...,N-1
  AND B[0:N-1], C[1:N-1] ARE THE RECURRENCE COEFFICIENTS AND
  P[1](X) = X - B[0], P[0](X) = 1 (SEE STOER, 192, PAR. 3.5, P. 119).
  THE ALGORITHM MAY BE DESCRIBED BY

```

$$\begin{array}{c} / R[K] \backslash \\ [ \quad ] (X) \\ \backslash S[K] / \end{array} = \begin{array}{c} / 0 \quad -C[K] \backslash \\ [ \quad ] \\ \backslash 1 \quad X-B[K] / \end{array} * \begin{array}{c} / R[K+1] \backslash \\ [ \quad ] (X) \\ \backslash S[K+1] / \end{array} + \begin{array}{c} / A[K-1] \backslash \\ [ \quad ] \\ \backslash 0 / \end{array}$$

```

FOR K=N-1,N-2,...,1, WITH INITIAL VALUES R[N]=A[N-1],S[N]=A[N].
THEN

```

$$\text{SERORTPOL} = ( P[0](X), P[1](X) ) * \begin{array}{c} / R[1] \backslash \\ [ \quad ] (X) \\ \backslash S[1] / \end{array}$$

(SEE ALSO LUKE, 1969, PAR. 8.7, P. 327).

THE RECURRENCE COEFFICIENTS MAY BE OBTAINED BY USE OF THE PROCEDURE RECCOF OR ADAPTING IN ABRAMOWITZ AND STEGUN, 1964, TABLE 22.7, P. 782 AS FOLLOWS: (NOTE THAT IN THAT TABLE  $N \geq 1$ , FOR  $N=0$  CONSULT 22.4)

```

B[I]:=A2(I)/A3(I) , I=0,1,2,...,N-1
C[I]:=A4(I)/(A3(I)*A3(I-1))*A1(I-1) , I=1,2,...,N-1

```

```

"IF" N=0 "THEN" SERORTPOL:=A[0]
"ELSE"
"BEGIN" "INTEGER" K; "REAL" H,R,S;
R:=A[N-1]; S:=A[N];
"FOR" K=N-1 "STEP" =1 "UNTIL" 1 "DO"
"BEGIN"
H:=R;
R:=A[K-1]-C[K]*S;
S:=H+(X-B[K])*S;
"END";
SERORTPOL:=R+(X-B[0])*S
"END" SERORTPOL;
"EOP"

```



SECTION : 2.2.2.2

(OCTOBER 1975)

PAGE 1

AUTHOR : C.G. VAN DER LAAN.

INSTITUTE : RIJKSUNIVERSITEIT GRONINGEN.

RECEIVED : 740131.

BRIEF DESCRIPTION :

THIS SECTION CONTAINS THREE PROCEDURES;  
CHEPOLSER EVALUATES A CHEBYSHEV SERIES;  
CHEPOL EVALUATES A CHEBYSHEV POLYNOMIAL;  
ALLCHEPOL EVALUATES ALL CHEBYSHEV POLYNOMIALS LOWER THAN A  
CERTAIN DEGREE.

KEYWORDS :

CHEBYSHEV SERIES EVALUATION,  
GOERTZEL, WATT, CLENSHAW ALGORITHM.  
CHEBYSHEV POLYNOMIAL EVALUATION,  
LINEAR THREE TERM RECURRENCE RELATION.

SUBSECTION: CHEPOLSER.

CALLING SEQUENCE :

THE HEADING OF THE PROCEDURE READS :  
"REAL""PROCEDURE"CHEPOLSER(N,X,A);  
"VALUE"N,X;"INTEGER"N;"REAL"X;"ARRAY"A;

CHEPOLSER:=THE VALUE OF THE CHEBYSHEV SERIES  
 $A[0]+A[1]*T_1(X)+\dots+A[N]*T_N(X)$ ,  
WHERE  $T_1(X), \dots, T_N(X)$  ARE CHEBYSHEV POLYNOMIALS  
OF THE FIRST KIND, OF DEGREE 1,  $\dots, N$ , RESPECTIVELY.

THE MEANING OF THE FORMAL PARAMETERS IS :

N: <ARITHMETIC EXPRESSION>;

ENTRY:

THE DEGREE OF THE POLYNOMIAL REPRESENTED BY THE CHEBYSHEV  
SERIES;

X: <ARITHMETIC EXPRESSION>;

ENTRY:

THE ARGUMENT OF THE CHEBYSHEV POLYNOMIALS;

A: <ARRAY IDENTIFIER>;

"ARRAY" A[0:N];

ENTRY:

THE COEFFICIENTS OF THE CHEBYSHEV SERIES MUST BE GIVEN IN  
ARRAY A, WHERE A[I] IS COEFFICIENT OF THE CHEBYSHEV POLYNOMIAL OF  
DEGREE I,  $0 \leq I \leq N$ .

PROCEDURES USED: NONE.



SECTION : 2.2.2.2

(OCTOBER 1975)

PAGE 2

RUNNING TIME : PROPORTIONAL TO N.

LANGUAGE : ALGOL 60.

METHOD AND PERFORMANCE :

FOR A DESCRIPTION OF THE ALGORITHM SEE STOER, 1972, P. 62.  
 AN UPPER BOUND FOR THE (ABSOLUTE) ERROR IS A QUADRATIC FUNCTION OF  
 THE DEGREE OF THE POLYNOMIAL REPRESENTED BY THE CHEBYSHEV SERIES.  
 (VAN DER LAAN).  
 THIS UPPER BOUND IS A ROUGH OVER-ESTIMATE FOR THE SPECIAL CASES  
 $ABS(X) < .5$  AND/OR A DESCENDING SERIES. (VAN DER LAAN).

REFERENCES : SEE ALLCHEPOL (THIS SECTION).

EXAMPLE OF USE :

AS A FORMAL TEST OF THE PROCEDURE THE POLYNOMIAL  
 $1 + 1/2 * T_1(X) + 1/4 * T_2(X)$  IS EVALUATED FOR  $X = -1, 0, 1$ , WHERE  $T_1(X)$  AND  
 $T_2(X)$  ARE CHEBYSHEV POLYNOMIALS OF THE FIRST AND SECOND DEGREE,  
 RESPECTIVELY.

```

"BEGIN""ARRAY"A[0:2];
  "REAL""PROCEDURE"CHEPOLSER(N,X,A);"CODE"31046;
  A[2]:= .25;A[1]:= .5;A[0]:= 1;
  OUTPUT(61,"("3(BZ.DD)""",CHEPOLSER(2,-1,A),CHEPOLSER(2,0,A),
  CHEPOLSER(2,1,A))
"END"

```

.75 .75 1.75

SUBSECTION : CHEPOL.

CALLING SEQUENCE :

THE HEADING OF THE PROCEDURE READS:  
 "REAL""PROCEDURE"CHEPOL(N,X);  
 "VALUE"N,X;"INTEGER"N;"REAL"X;

CHEPOL:=THE VALUE OF THE CHEBYSHEV POLYNOMIAL OF THE FIRST KIND OF  
 DEGREE N FOR THE ARGUMENT X.

THE MEANING OF THE FORMAL PARAMETERS IS:

N : &lt;ARITHMETIC EXPRESSION&gt;;

ENTRY:

THE DEGREE OF THE POLYNOMIAL ( $\geq 0$ );

X : &lt;ARITHMETIC EXPRESSION&gt;;

ENTRY:

THE ARGUMENT OF THE CHEBYSHEV POLYNOMIAL.



SECTION : 2.2.2.2

(OCTOBER 1975)

PAGE 3

PROCEDURES USED : NONE.

RUNNING TIME : PROPORTIONAL TO N.

LANGUAGE : ALGOL 60.

METHOD AND PERFORMANCE : SEE ALLCHEPOL (THIS SECTION).

REFERENCES : SEE ALLCHEPOL (THIS SECTION).

EXAMPLE OF USE :

SUBSECTION : ALLCHEPOL.

CALLING SEQUENCE :

THE HEADING OF THE PROCEDURE READS :  
"PROCEDURE"ALLCHEPOL(N,X,T);  
"VALUE"N,X;"INTEGER"N;"REAL"X;"REAL""ARRAY"T;

THE MEANING OF THE FORMAL PARAMETERS IS:

N : <ARITHMETIC EXPRESSION>;

ENTRY:

THE DEGREE OF THE LAST POLYNOMIAL ( $\geq 0$ );

X : <ARITHMETIC EXPRESSION>;

ENTRY:

THE ARGUMENT OF THE CHEBYSHEV POLYNOMIALS.

T : <ARRAY IDENTIFIER>;

"ARRAY" T[0:N];

EXIT:

THE VALUES OF THE CHEBYSHEV POLYNOMIALS OF THE FIRST KIND OF DEGREES 0,1,...,N, FOR THE ARGUMENT X, ARE DELIVERED IN T[0],T[1],...,T[N], RESPECTIVELY.

PROCEDURES USED: NONE.

RUNNING TIME : PROPORTIONAL TO N.

LANGUAGE : ALGOL 60.



## METHOD AND PERFORMANCE:

FOR A DESCRIPTION OF THE ALGORITHM SEE STOER, 1972, P. 21.  
 USE HAS BEEN MADE OF THE FOLLOWING NORMALIZATION: THE MAXIMUM  
 (ABSOLUTE) VALUE OF THE CHEBYSHEV POLYNOMIAL EQUALS 1.  
 (THE NORMALIZATION 'LEADING COEFFICIENT=1' WOULD LEAD TO UNDERFLOW  
 FOR  $N=974$  (APPROXIMATELY, DEPENDING ON  $X$ ) ON THE CDC 6000 SYSTEMS).  
 AN UPPER BOUND FOR THE (ABSOLUTE) ERROR IS A QUADRATIC FUNCTION  
 OF THE DEGREE OF THE CHEBYSHEV POLYNOMIAL. THIS UPPER BOUND IS A  
 ROUGH OVER-ESTIMATE FOR THE SPECIAL CASE  $ABS(X) < .5$  (STOER, 1972,  
 P. 21-24 & VAN DER LAAN).

## REFERENCES :

LAAN, C. G. VAN DER (TO APPEAR) :  
 ORTHOGONAL POLYNOMIALS IN NUMERICAL ANALYSIS 1.  
 ERROR ANALYSIS OF LINEAR TWO TERM AND THREE TERM RECURRENCE  
 RELATIONS.

STOER, J. (1972) :  
 EINFUEHRUNG IN DIE NUMERISCHE MATHEMATIK 1.  
 HEIDELBERGER TASCHENBUECHER 105, SPRINGER-VERLAG.

## EXAMPLE OF USE :

AS A FORMAL TEST OF THE PROCEDURE ALLCHEPOL THE CHEBYSHEV  
 POLYNOMIALS OF THE FIRST KIND OF DEGREES 0, 1, 2 ARE EVALUATED FOR  
 THE ARGUMENTS -1, 0, 1.

```
"BEGIN" "ARRAY" T(0:2); "PROCEDURE" ALLCHEPOL (N, X, T); "CODE" 31043;
  ALLCHEPOL (2, -1, T); OUTPUT (61, (" /, 3 (-DB) ")", T(0), T(1), T(2));
  ALLCHEPOL (2, 0, T); OUTPUT (61, (" /, 3 (-DB) ")", T(0), T(1), T(2));
  ALLCHEPOL (2, 1, T); OUTPUT (61, (" /, 3 (-DB) ")", T(0), T(1), T(2));
"END"
```

```
1 -1 1
1 0 -1
1 1 1
```



SECTION : 2.2.2.2

(OCTOBER 1975)

PAGE 5

## SOURCE TEXT(S) :

```

"CODE"31046;
"REAL" "PROCEDURE" CHEPOLSER(N,X,A);
"VALUE" N,X;"INTEGER" N;"REAL" X;"ARRAY" A;
"IF" N=0 "THEN" CHEPOLSER:=A[0] "ELSE"
"IF" N=1 "THEN" CHEPOLSER:=A[0]+A[1]*X "ELSE"
"BEGIN" "INTEGER" K;"REAL" H,R,S,TX;
  TX:=X+X;R:=A[N];
  H:=A[N-1]+R*TX;
  "FOR" K:=N-2 "STEP" -1 "UNTIL" 1 "DO"
  "BEGIN" S:=R;R:=H;
    H:=A[K]+R*TX-S
  "END"K;
  CHEPOLSER:=A[0]-R+H*X
"END" CHEPOLSER;
  "EOP"

```

```

"CODE"31042;
"REAL""PROCEDURE"CHEPOL(N,X);"VALUE"N,X;"INTEGER"N;"REAL"X;
"IF" N = 0 "THEN" CHEPOL :=1 "ELSE"
"IF" N = 1 "THEN" CHEPOL :=X "ELSE"
"BEGIN""INTEGER" I;"REAL" T1,T2,H,X2;
  T2:=X;T1:=1;X2:=X+X;
  "FOR" I:=2 "STEP" 1 "UNTIL" N "DO"
  "BEGIN" H:=X2*T2-T1;T1:=T2;T2:=H"END";
  CHEPOL:=H
"END"CHEPOL;
  "EOP"

```

```

"CODE"31043;
"PROCEDURE"ALLCHEPOL(N,X ,T);
"VALUE"N,X ;"INTEGER"N;"REAL"X ;"REAL""ARRAY"T;
"IF" N = 0 "THEN" T[0] :=1 "ELSE"
"IF" N = 1 "THEN" "BEGIN" T[0] := 1; T[1] := X "END" "ELSE"
"BEGIN""INTEGER" I;"REAL" T1,T2,H,X2;
  T[0]:=T1:=1;T[1]:=T2:=X;X2:=X+X;
  "FOR" I:=2 "STEP" 1 "UNTIL" N "DO"
  "BEGIN" T[I] := H:=X2*T2-T1;T1:=T2;T2:=H"END"
"END"ALLCHEPOL;
  "EOP"

```



SECTION : 2.2.3.1

(OCTOBER 1974)

PAGE 1

AUTHOR: C.G. VAN DER LAAN.

INSTITUTE: RIJKSUNIVERSITEIT GRONINGEN.

RECEIVED: 740701.

BRIEF DESCRIPTION:

THIS SECTION CONTAINS THE PROCEDURES:  
 SINSER FOR EVALUATING A SINE SERIES;  
 COSSER FOR EVALUATING A COSINE SERIES;  
 FOUUSER, FOUUSER1, FOUUSER2 FOR EVALUATING A FOURIER SERIES  
 (IN FOUUSER THE SERIES IS RESTRICTED TO A SERIES WITH SINE  
 COEFFICIENTS EQUAL TO COSINE COEFFICIENTS);  
 COMFOUSER, COMFOUSER1, COMFOUSER2 FOR EVALUATING A COMPLEX FOURIER  
 SERIES  
 (IN COMFOUSER THE SERIES IS RESTRICTED TO A SERIES WITH REAL  
 COEFFICIENTS).

KEYWORDS:

FINITE FOURIER SERIES EVALUATION,  
 TRIGONOMETRIC POLYNOMIAL EVALUATION,  
 GOERTZEL, WATT, CLENSHAW, REINSCH ALGORITHM,  
 LINEAR THREE-TERM INHOMOGENEOUS RECURRENCE RELATION.



SECTION : 2.2.3.1

(OCTOBER 1974)

PAGE 2

SUBSECTION : SINSEB.

CALLING SEQUENCE:

THE HEADING OF THE PROCEDURE READS:  
 "REAL" "PROCEDURE" SINSEB(N, THETA, B);  
 "VALUE" N, THETA; "INTEGER" N; "REAL" THETA; "ARRAY" B;

SINSEB: THE VALUE OF THE SINE SERIES  
 $B[1] * \sin(\text{THETA}) + \dots + B[N] * \sin(N * \text{THETA})$ .

THE MEANING OF THE FORMAL PARAMETERS IS:  
 N: <ARITHMETIC EXPRESSION>;  
 ENTRY: THE NUMBER OF TERMS IN THE SINE SERIES;  
 THETA: <ARITHMETIC EXPRESSION>;  
 ENTRY: THE ARGUMENT OF THE SINE SERIES;  
 B: <ARRAY IDENTIFIER>;  
 "ARRAY" B[1:N];  
 ENTRY: THE COEFFICIENTS OF THE SINE SERIES.

PROCEDURES USED: NONE.

RUNNING TIME: PROPORTIONAL TO N  
 (IN FIRST ORDER; N MULTIPLICATIONS; 3N ADDITIONS;  
 3 SINE/COSINE EVALUATIONS).

LANGUAGE: ALGOL 60.

METHOD AND PERFORMANCE : SEE COMFOUSER2 (THIS SECTION).



SECTION : 2.2.3.1

(OCTOBER 1974)

PAGE 3

SUBSECTION : COSSER.

CALLING SEQUENCE:

THE HEADING OF THE PROCEDURE READS;  
 "REAL" "PROCEDURE" COSSER(N, THETA, A);  
 "VALUE" N, THETA; "INTEGER" N; "REAL" THETA; "ARRAY" A;

COSSER := THE VALUE OF THE COSINE SERIES  
 $A[0] + A[1] * \cos(\text{THETA}) + \dots + A[N] * \cos(N * \text{THETA})$ .

THE MEANING OF THE FORMAL PARAMETERS IS:

N: <ARITHMETIC EXPRESSION>;  
 ENTRY: THE DEGREE OF THE TRIGONOMETRIC POLYNOMIAL.

THETA: <ARITHMETIC EXPRESSION>;  
 ENTRY: THE ARGUMENT OF THE COSINE SERIES.

A: <ARRAY IDENTIFIER>;  
 "ARRAY" A[0:N];  
 ENTRY: THE COEFFICIENTS OF THE COSINE SERIES.

PROCEDURES USED: NONE.

RUNNING TIME: PROPORTIONAL TO N  
 (IN FIRST ORDER: N MULTIPLICATIONS; 3N ADDITIONS;  
 2 COSINE/SINE EVALUATIONS).

LANGUAGE: ALGOL 60.

METHOD AND PERFORMANCE : SEE COMFOUSER2 (THIS SECTION).



SECTION : 2.2.3.1

(OCTOBER 1974)

PAGE 4

SUBSECTION : FOUUSER.

CALLING SEQUENCE:

THE HEADING OF THE PROCEDURE READS:  
 "REAL" "PROCEDURE" FOUUSER (N, THETA, A);  
 "VALUE" N, THETA; "INTEGER" N; "REAL" THETA; "ARRAY" A;

FOUSER := THE VALUE OF THE FOURIER SERIES  
 $A[0] + A[1] * (\cos(\text{THETA}) + \sin(\text{THETA})) + \dots + A[N] * (\cos(N * \text{THETA}) + \sin(N * \text{THETA}))$ .

THE MEANING OF THE FORMAL PARAMETERS IS:  
 N: <ARITHMETIC EXPRESSION>;  
 ENTRY: THE DEGREE OF THE TRIGONOMETRIC POLYNOMIAL;  
 THETA: <ARITHMETIC EXPRESSION>;  
 ENTRY: THE ARGUMENT OF THE FOURIER SERIES;  
 A: <ARRAY IDENTIFIER>;  
 "ARRAY" A[0:N];  
 ENTRY: THE COEFFICIENTS OF THE (FINITE) FOURIER SERIES.

PROCEDURES USED: NONE.

RUNNING TIME: PROPORTIONAL TO N  
 (IN FIRST ORDER; N MULTIPLICATIONS; 3N ADDITIONS;  
 3 COSINE/SINE EVALUATIONS).

LANGUAGE: ALGOL 60.

METHOD AND PERFORMANCE : SEE COMFOUSER2 (THIS SECTION).



SECTION : 2.2.3.1

(OCTOBER 1974)

PAGE 5

SUBSECTION : FOUER1.

CALLING SEQUENCE:

THE HEADING OF THE PROCEDURE READS:  
 "REAL" "PROCEDURE" FOUER1(N, THETA, A, B);  
 "VALUE" N, THETA; "INTEGER" N; "REAL" THETA; "ARRAY" A, B;

FOUER1: = THE VALUE OF THE FOURIER SERIES  
 $A[0] + A[1] * \cos(\text{THETA}) + B[1] * \sin(\text{THETA}) + \dots$   
 $+ A[N] * \cos(N * \text{THETA}) + B[N] * \sin(N * \text{THETA}).$

THE MEANING OF THE FORMAL PARAMETERS IS:

N: <ARITHMETIC EXPRESSION>;  
 ENTRY: THE DEGREE OF THE TRIGONOMETRIC POLYNOMIAL;  
 THETA: <ARITHMETIC EXPRESSION>;  
 ENTRY: THE ARGUMENT OF THE FOURIER SERIES;  
 A, B: <ARRAY IDENTIFIER>;  
 "ARRAY" A[0:N], B[1:N];  
 ENTRY: THE COEFFICIENTS OF THE (FINITE) FOURIER SERIES,  
 WITH A[K] COEFFICIENT OF  $\cos(K * \text{THETA})$ , (K=0, ..., N)  
 AND B[K] COEFFICIENT OF  $\sin(K * \text{THETA})$ , (K=1, ..., N).

PROCEDURES USED: NONE.

RUNNING TIME: PROPORTIONAL TO N  
 (IN FIRST ORDER; 4N MULTIPLICATIONS; 4N ADDITIONS;  
 2 COSINE/SINE EVALUATIONS).

LANGUAGE: ALGOL 60.

METHOD AND PERFORMANCE : SEE COMFOUER2 (THIS SECTION).



SUBSECTION : FOUUSER2.

CALLING SEQUENCE:

THE HEADING OF THE PROCEDURE READS:  
 "REAL" "PROCEDURE" FOUUSER2(N, THETA, A, B);  
 "VALUE" N, THETA; "INTEGER" N; "REAL" THETA; "ARRAY" A, B;

FOUSER2: = THE VALUE OF THE FOURIER SERIES  
 $A[0] + A[1] * \cos(\text{THETA}) + B[1] * \sin(\text{THETA}) + \dots$   
 $+ A[N] * \cos(N * \text{THETA}) + B[N] * \sin(N * \text{THETA}).$

THE MEANING OF THE FORMAL PARAMETERS IS:  
 N: <ARITHMETIC EXPRESSION>;  
 ENTRY: THE DEGREE OF THE TRIGONOMETRIC POLYNOMIAL;  
 THETA: <ARITHMETIC EXPRESSION>;  
 ENTRY: THE ARGUMENT OF THE FOURIER SERIES;  
 A, B: <ARRAY IDENTIFIER>;  
 "ARRAY" A[0:N], B[1:N];  
 ENTRY: THE COEFFICIENTS OF THE (FINITE) FOURIER SERIES,  
 WITH A[K] COEFFICIENT OF  $\cos(K * \text{THETA})$ , (K=0, ..., N)  
 AND B[K] COEFFICIENT OF  $\sin(K * \text{THETA})$ , (K=1, ..., N).

PROCEDURES USED: SINSER = CP31090,  
 COSSER = CP31091.

RUNNING TIME: PROPORTIONAL TO N  
 (IN FIRST ORDER: 2N MULTIPLICATIONS; 6N ADDITIONS;  
 6 COSINE/SINE EVALUATIONS).

LANGUAGE : ALGOL 60.

METHOD AND PERFORMANCE : SEE COMFOUSER2 (THIS SECTION).



SECTION : 2.2.3.1

(OCTOBER 1974)

PAGE 7

SUBSECTION : COMFOUSER.

CALLING SEQUENCE:

THE HEADING OF THE PROCEDURE READS:  
 "PROCEDURE"COMFOUSER (N,THETA,A,RR,RI);  
 "VALUE"N,THETA;"INTEGER"N;"REAL"THETA,RR,RI;"ARRAY"A;

THE MEANING OF THE FORMAL PARAMETERS IS:

N: <ARITHMETIC EXPRESSION>;  
 ENTRY: THE DEGREE OF THE POLYNOMIAL IN  $\exp(i\theta)$ ;  
 THETA: <ARITHMETIC EXPRESSION>;  
 ENTRY: THE ARGUMENT OF THE FOURIER SERIES;  
 A: <ARRAY IDENTIFIER>;  
 "ARRAY"A[0:N];  
 ENTRY: THE REAL COEFFICIENTS A[K] ( $k=0, \dots, N$ ) IN THE SERIES  

$$f_n(\theta) = A[0] + A[1] * \exp(i\theta) + \dots + A[N] * \exp(i\theta * N)$$
 MUST BE GIVEN IN ARRAY A;  
 RR,RI: <VARIABLE>;  
 EXIT: THE REAL PART AND THE IMAGINARY PART OF  $f_n(\theta)$   
 ARE DELIVERED IN RR AND RI, RESPECTIVELY.

PROCEDURES USED: NONE.

RUNNING TIME: PROPORTIONAL TO N  
 (IN FIRST ORDER: N MULTIPLICATIONS; 3N ADDITIONS;  
 3 COSINE/SINE EVALUATIONS).

LANGUAGE: ALGOL 60.

METHOD AND PERFORMANCE : SEE COMFOUSER2 (THIS SECTION).



SECTION : 2,2.3,1

(OCTOBER 1974)

PAGE 8

SUBSECTION : COMFOUSER1.

CALLING SEQUENCE:

THE HEADING OF THE PROCEDURE READS:  
 "PROCEDURE"COMFOUSER1(N,THETA,AR,AI,RR,RI);  
 "VALUE"N,THETA;"INTEGER"N;"REAL"THETA,RR,RI;"ARRAY"AR,AI;

THE MEANING OF THE FORMAL PARAMETERS IS:

N: <ARITHMETIC EXPRESSION>;  
 ENTRY: THE DEGREE OF THE POLYNOMIAL IN  $\exp(i\theta)$ ;  
 THETA: <ARITHMETIC EXPRESSION>;  
 ENTRY: THE ARGUMENT OF THE FOURIER SERIES;  
 AR,AI: <ARRAY IDENTIFIER>;  
 "ARRAY"AR,AI[0:N];  
 ENTRY: THE REAL PART AND THE IMAGINARY PART OF THE COMPLEX  
 COEFFICIENTS  $C[K]$  ( $K=0, \dots, N$ ) IN THE SERIES  

$$FN(\theta) = C[0] + C[1] * \exp(i\theta) + \dots + C[N] * \exp(i\theta) * i^N$$
  
 MUST BE GIVEN IN ARRAY AR AND AI, RESPECTIVELY;  
 RR,RI: <VARIABLE>;  
 EXIT: THE REAL PART AND THE IMAGINARY PART OF  $FN(\theta)$   
 ARE DELIVERED IN RR AND RI, RESPECTIVELY.

PROCEDURES USED: NONE.

RUNNING TIME: PROPORTIONAL TO N  
 (IN FIRST ORDER: 4N MULTIPLICATIONS; 4N ADDITIONS;  
 2 COSINE/SINE EVALUATIONS).

LANGUAGE: ALGOL 60.

METHOD AND PERFORMANCE : SEE COMFOUSER2 (THIS SECTION).



SECTION : 2.2.3.1

(OCTOBER 1974)

PAGE 9

SUBSECTION : COMFOUSER2.

CALLING SEQUENCE:

THE HEADING OF THE PROCEDURE READS:  
 "PROCEDURE"COMFOUSER2(N,THETA,AR,AI,RR,RI);  
 "VALUE"N,THETA;"INTEGER"N;"REAL"THETA,RR,RI;"ARRAY"AR,AI;

THE MEANING OF THE FORMAL PARAMETERS IS:

N: <ARITHMETIC EXPRESSION>;  
 ENTRY: THE DEGREE OF THE POLYNOMIAL IN  $\exp(i\theta)$ ;  
 THETA: <ARITHMETIC EXPRESSION>;  
 ENTRY: THE ARGUMENT OF THE FOURIER SERIES;  
 AR,AI: <ARRAY IDENTIFIER>;  
 "ARRAY"AR,AI[0:N];  
 ENTRY: THE REAL PART AND THE IMAGINARY PART OF THE COMPLEX  
 COEFFICIENTS  $C[K]$  ( $K=0,\dots,N$ ) IN THE SERIES  

$$FN(\theta) = C[0] + C[1] * \exp(i\theta) + \dots + C[N] * \exp(i\theta) * i^N$$
  
 MUST BE GIVEN IN ARRAY AR AND AI, RESPECTIVELY;  
 RR,RI: <VARIABLE>;  
 EXIT: THE REAL PART AND THE IMAGINARY PART OF  $FN(\theta)$   
 ARE DELIVERED IN RR AND RI, RESPECTIVELY.

PROCEDURES USED: COMFOUSER= CP31095.

RUNNING TIME: PROPORTIONAL TO N  
 (IN FIRST ORDER; 2N MULTIPLICATIONS; 6N ADDITIONS;  
 6 COSINE/SINE EVALUATIONS).

LANGUAGE: ALGOL 60.

METHOD AND PERFORMANCE:

FOR THE EVALUATION OF A FINITE FOURIER SERIES  
 (=TRIGONOMETRIC POLYNOMIAL OF DEGREE N SEE POLYA AND SZEGOE, 1971,  
 P. 76)

$$FN(\theta) = A[0] + A[1] * \cos(\theta) + B[1] * \sin(\theta) + \dots +$$

$$A[N] * \cos(N\theta) + B[N] * \sin(N\theta),$$

TWO ALGORITHMS ARE USED:



1. HORNER SCHEME  
 LET  $C[K] = A[K] + I * B[K]$ ,  $K = 0, \dots, N$   
 AND  $Z = \text{EXP}(-I * \text{THETA})$   
 THEN  

$$FN(\text{THETA}) = \text{RE}(C[0] + C[1] * Z + \dots + C[N] * Z^{**N}).$$
 THE ALGORITHM IS GIVEN BY:  
 $P := C[N]$   
 $P := P * Z + C[K]$ ,  $K = N-1, \dots, 0$   
 $FN(\text{THETA}) := \text{RE}(P).$   
 (FOUSER1)
  2. A COMBINATION OF THE CLENSHAW ALGORITHM (SEE GENTLEMAN(1969,II)  
 , VAN DER LAAN, LUKE(1969, P.327-329) OR STOER(1972, P.62,63))  
 AND THE MODIFICATION OF REINSCH (SEE REINSCH(1967), VAN DER  
 LAAN, STOER(1972, P.64,65)).  
 (SINSER, COSSER, FOUUSER, FOUUSER2)
- A MODIFICATION OF THE IDEA OF NEWBERY IS NOT IMPLEMENTED BECAUSE  
 OF THE INTRODUCTION OF SINE (COSINE) TERMS IN A COSINE (SINE)  
 SERIES AND THE THE INEFFICIENCY OF THE ALGORITHM (SEE VAN DER  
 LAAN OR NEWBERY(1973)).

FOR THE EVALUATION OF A FINITE COMPLEX FOURIER SERIES  

$$FN(\text{THETA}) = AR[0] + I * AI[0] + (AR[1] + I * AI[1]) * \text{EXP}(I * \text{THETA}) + \dots$$

$$+ (AR[N] + I * AI[N]) * \text{EXP}(I * \text{THETA})^{**N},$$

TWO ALGORITHMS, IN REAL ARITHMETIC, ARE USED:

1. HORNER SCHEME  
 LET  $C[K] = AR[K] + I * AI[K]$ ,  $K = 0, \dots, N$   
 AND  $Z = \text{EXP}(I * \text{THETA})$   
 THEN  

$$FN(\text{THETA}) = C[0] + C[1] * Z + \dots + C[N] * Z^{**N}.$$
 THE ALGORITHM IS GIVEN BY  
 $P := C[N]$   
 $P := P * Z + C[K]$ ,  $K = N-1, N-2, \dots, 0$   
 $FN(\text{THETA}) := P.$   
 (COMFOUSER1)
  2. A COMBINATION OF THE CLENSHAW ALGORITHM AND THE MODIFICATION OF  
 REINSCH,  
 LET  $CAR = AR[0] + AR[1] * \text{COS}(\text{THETA}) + \dots + AR[N] * \text{COS}(N * \text{THETA}),$   
 $SAI = AI[1] * \text{SIN}(\text{THETA}) + \dots + AI[N] * \text{SIN}(N * \text{THETA}),$   
 $SAR = AR[1] * \text{SIN}(\text{THETA}) + \dots + AR[N] * \text{SIN}(N * \text{THETA}),$   
 $CAI = AI[0] + AI[1] * \text{COS}(\text{THETA}) + \dots + AI[N] * \text{COS}(N * \text{THETA})$   
 THEN  $FN(\text{THETA}) = CAR - SAI + I * (SAR + CAI).$   
 (COMFOUSER, COMFOUSER2)
- THE HORNER SCHEME IS IMPLEMENTED BECAUSE OF THE SIMPLICITY OF  
 THE ALGORITHM (ALTHOUGH THIS ALGORITHM IS LESS EFFICIENT THAN THE  
 GOERTZEL/WATT/CLENSHAW/REINSCH ALGORITHM) AND THE STABLE NATURE  
 OF ORTHOGONAL TRANSFORMATIONS,  
 A COMBINATION OF THE ALGORITHM OF GOERTZEL/WATT/CLENSHAW AND THE  
 MODIFICATION OF REINSCH IS IMPLEMENTED BECAUSE OF THE EFFICIENCY  
 OF THE GWC ALGORITHM AND THE STABILITY OF THE MODIFICATION OF  
 REINSCH, ESPECIALLY FOR SMALL VALUES OF THE ARGUMENT (MOD. PI).  
 AN UPPER BOUND FOR THE ERROR GROWTH IS GIVEN BY A LINEAR FUNCTION  
 OF THE DEGREE FOR BOTH (IMPLEMENTED) ALGORITHMS (SEE VAN DER LAAN).



## REFERENCES:

GENTLEMAN, W. M. (1969):  
 AN ERROR ANALYSIS OF GOERTZEL'S (WATT'S) METHOD FOR COMPUTING  
 FOURIER COEFFICIENTS.  
 COMP. J., VOL. 12, P. 160-165.

LAAN, C. G. VAN DER (TO APPEAR):  
 ORTHOGONAL POLYNOMIALS IN NUMERICAL ANALYSIS I.  
 ERROR ANALYSIS OF LINEAR TWO-TERM AND THREE-TERM RECURRENCE  
 RELATIONS.

LUKE, Y. L. (1969):  
 THE SPECIAL FUNCTIONS AND THEIR APPROXIMATIONS, VOL. 1.  
 ACADEMIC PRESS.

NEWBERY, A. C. R. (1973):  
 ERROR ANALYSIS FOR FOURIER SERIES EVALUATION.  
 MATH. COMP., VOL. 26, P. 923-924.

POLYA, G. AND G. SZEGOE (1971):  
 AUFGABEN UND LEHRSAETZE AUS DER ANALYSIS II.  
 HEIDELBERGER TASCHENBUECHER 74, SPRINGER.

REINSCH, C. (1967):  
 A NOTE ON TRIGONOMETRIC INTERPOLATION.  
 BERICHT NR. 6709.  
 ABTEILUNG MATHEMATIK DER TECHNISCHEN UNIVERSITAET MUENCHEN.

STOER, J. (1972):  
 EINFUEHRUNG IN DIE NUMERISCHE MATHEMATIK I.  
 HEIDELBERGER TASCHENBUECHER 105, SPRINGER.

## EXAMPLE OF USE:

THE FOURIER SERIES  $.5 + \cos(\theta) + \sin(\theta)$   
 IS EVALUATED FOR THE ARGUMENTS  $0, \pi/2, \pi$ , BY MEANS OF FOUUSER

```
"BEGIN" "REAL" THETA, PI, "ARRAY" A [0:1];
  "REAL" "PROCEDURE" FOUUSER(N, THETA, A ); "CODE" 31092;
  PI := ARCTAN(1) * 4; A [0] := .5; A [1] := 1;
  "FOR" THETA := 0, PI/2, PI "DO"
  OUTPUT(61, "( /, B=D, DD)", FOUUSER(1, THETA, A))
"END"
```

```
1.50
1.50
-0.50
```



## SOURCE TEXTS:

```

"CODE" 31090;
"REAL" "PROCEDURE" SINSER(N, THETA, B);
"VALUE" N, THETA; "INTEGER" N; "REAL" THETA; "ARRAY" B;
"BEGIN" "INTEGER" K; "REAL" C, CC, LAMBDA, H, DUN, UN, UN1;
    C := COS(THETA);
    "IF" C <= .5 "THEN"
        "BEGIN" LAMBDA := 4 * COS(THETA/2) ** 2; UN := DUN := 0;
            "FOR" K := N "STEP" -1 "UNTIL" 1 "DO"
                "BEGIN" DUN := LAMBDA * UN - DUN + B[K] ;
                    UN := DUN - UN;
                "END"
            "END" "ELSE" "IF" C > .5 "THEN"
                "BEGIN" LAMBDA := -4 * SIN(THETA/2) ** 2; UN := DUN := 0;
                    "FOR" K := N "STEP" -1 "UNTIL" 1 "DO"
                        "BEGIN" DUN := LAMBDA * UN + DUN + B[K] ;
                            UN := DUN + UN;
                        "END"
                    "END" "ELSE"
                        "BEGIN" CC := C + C; UN1 := UN1 := 0;
                            "FOR" K := N "STEP" -1 "UNTIL" 1 "DO"
                                "BEGIN" H := CC * UN - UN1 + B[K]; UN1 := UN; UN := H; "END"
                            "END";
                        SINSER := UN * SIN(THETA)
                    "END" SINSER;
                    "EOP"

```

```

"CODE" 31091;
"REAL" "PROCEDURE" COSSER(N, THETA, A);
"VALUE" N, THETA; "INTEGER" N; "REAL" THETA; "ARRAY" A;
"BEGIN" "INTEGER" K; "REAL" C, CC, LAMBDA, H, DUN, UN, UN1;
    C := COS(THETA);
    "IF" C <= .5 "THEN"
        "BEGIN" LAMBDA := 4 * COS(THETA/2) ** 2; UN := DUN := 0;
            "FOR" K := N "STEP" -1 "UNTIL" 0 "DO"
                "BEGIN" UN := DUN - UN;
                    DUN := LAMBDA * UN - DUN + A[K]
                "END"; COSSER := DUN - LAMBDA/2 * UN
            "END" "ELSE" "IF" C > .5 "THEN"
                "BEGIN" LAMBDA := -4 * SIN(THETA/2) ** 2; UN := DUN := 0;
                    "FOR" K := N "STEP" -1 "UNTIL" 0 "DO"
                        "BEGIN" UN := DUN + UN;
                            DUN := LAMBDA * UN + DUN + A[K]
                        "END"; COSSER := DUN - LAMBDA/2 * UN
                    "END" "ELSE"
                        "BEGIN" CC := C + C; UN1 := UN1 := 0;
                            "FOR" K := N "STEP" -1 "UNTIL" 1 "DO"
                                "BEGIN" H := CC * UN - UN1 + A[K];
                                    UN1 := UN; UN := H
                                "END"; COSSER := A[0] + UN * C - UN1
                            "END"
                        "END" COSSER;
                        "EOP"

```



```

"CODE" 31092;
"REAL" "PROCEDURE" FOUUSER (N, THETA, A);
"VALUE" N, THETA; "INTEGER" N; "REAL" THETA; "ARRAY" A;
"BEGIN" "INTEGER" K; "REAL" C, CC, LAMBDA, H, DUN, UN, UN1, C2, S2;
  C:=COS(THETA);
  "IF" C<=.5 "THEN"
    "BEGIN" C2:=COS(THETA/2); LAMBDA:=4*C2**2; UN:=DUN:=0;
      "FOR" K:=N"STEP"-1 "UNTIL" 0 "DO"
        "BEGIN" UN:=DUN-UN;
          DUN:=LAMBDA*UN-DUN+A[K]
        "END"; FOUUSER :=DUN+2*C2*(SIN(THETA/2)-C2)*UN
      "END" "ELSE" "IF" C>.5 "THEN"
        "BEGIN" S2:=SIN(THETA/2); LAMBDA:=-4*S2*S2; UN:=DUN:=0;
          "FOR" K:=N"STEP"-1 "UNTIL" 0 "DO"
            "BEGIN" UN:=DUN+UN;
              DUN:=LAMBDA*UN+DUN+A[K]
            "END"; FOUUSER :=DUN+2*S2*(S2+COS(THETA/2))*UN
          "END" "ELSE"
            "BEGIN" CC:=C+C; UN:=UN1:=0;
              "FOR" K:=N"STEP"-1 "UNTIL" 1 "DO"
                "BEGIN" H:=CC*UN-UN1+A[K];
                  UN1:=UN; UN:=H
                "END"; FOUUSER :=A[0]-UN1+(C+SIN(THETA))*UN
              "END"
            "END" FOUUSER;
            "EOP"

```

```

"CODE" 31093;
"REAL" "PROCEDURE" FOUUSER1 (N, THETA, A, B);
"VALUE" N, THETA; "INTEGER" N; "REAL" THETA; "ARRAY" A, B;
"BEGIN" "INTEGER" I; "REAL" R, S, H, CO, SI;
  R:=S:=0; CO:=COS(THETA); SI:=SIN(THETA);
  "FOR" I:=N"STEP"-1 "UNTIL" 1 "DO"
    "BEGIN" H:=CO*R+SI*S+A[I];
      S:=CO*S-SI*R+B[I];
      R:=H
    "END"; FOUUSER1:=CO*R+SI*S+A[0]
  "END" FOUUSER1;
  "EOP"

```

```

"CODE" 31094;
"REAL" "PROCEDURE" FOUUSER2 (N, THETA, A, B);
"VALUE" N, THETA; "INTEGER" N; "REAL" THETA; "ARRAY" A, B;
"BEGIN"
  "REAL" "PROCEDURE" SINSER (N, THETA, B); "CODE" 31090;
  "REAL" "PROCEDURE" COSSER (N, THETA, A); "CODE" 31091;
  FOUUSER2:=COSSER(N, THETA, A)+SINSER(N, THETA, B);
"END" FOUUSER2;
"EOP"

```



```

"CODE" 31095;
"PROCEDURE"COMFOUSER(N,THETA,A,RR,RI);
"VALUE"N,THETA;"INTEGER"N;"REAL"THETA,RR,RI;"ARRAY"A;
"BEGIN"INTEGER"K;"REAL"C,CC,LAMBDA,H,DUN,UN,UN1;
  C:=COS(THETA);
  "IF"C<=.5"THEN
    "BEGIN"LAMBDA:= 4*COS(THETA/2)**2;UN:=DUN:=0;
      "FOR"K:=N"STEP"-1"UNTIL"0"DO"
        "BEGIN"UN:=DUN-UN;
          DUN:=LAMBDA*UN-DUN+A[K]
        "END";RR :=DUN-LAMBDA/2*UN
      "END"ELSE"IF"C>.5"THEN
        "BEGIN"LAMBDA:=4*SIN(THETA/2)**2;UN:=DUN:=0;
          "FOR"K:=N"STEP"-1"UNTIL"0"DO"
            "BEGIN"UN:=DUN+UN;
              DUN:=LAMBDA*UN+DUN+A[K]
            "END";RR :=DUN-LAMBDA/2*UN
          "END"ELSE"
            "BEGIN"CC:=C+C;UN1:=UN1:=0;
              "FOR"K:=N"STEP"-1"UNTIL"1"DO"
                "BEGIN"H:=CC*UN-UN1+A[K];
                  UN1:=UN;UN:=H
                "END";RR :=A[0]+UN*C-UN1
              "END";RI:=UN*SIN(THETA)
            "END"COMFOUSER;
              "EOP"

"CODE" 31096;
"PROCEDURE"COMFOUSER1(N,THETA,AR,AI,RR,RI);
"VALUE"N,THETA;"INTEGER"N;"REAL"THETA,RR,RI;"ARRAY"AR,AI;
"BEGIN"INTEGER"K;"REAL"H,HR,HI,CO,SI;
  HR:=HI:=0;CO:=COS(THETA);SI:=SIN(THETA);
  "FOR"K:=N"STEP"-1"UNTIL"1"DO"
    "BEGIN"H:=CO*HR-SI*HI+AR[K];
      HI:=CO*HI+SI*HR+AI[K];
      HR:=H
    "END";
    RR:=CO*HR-SI*HI+AR[0];
    RI:=CO*HI+SI*HR+AI[0]
  "END"COMFOUSER1;
    "EOP"

"CODE" 31097;
"PROCEDURE"COMFOUSER2(N,THETA,AR,AI,RR,RI);
"VALUE"N,THETA;"INTEGER"N;"REAL"THETA,RR,RI;"ARRAY"AR,AI;
"BEGIN"REAL"CAR,CAI,SAR,SAI;
  "PROCEDURE"COMFOUSER(N,THETA,A,RR,RI);"CODE" 31095;
  COMFOUSER(N,THETA,AR,CAR,SAR);
  COMFOUSER(N,THETA,AI,CAI,SAI);
  RR:=CAR-SAI;
  RI:=CAI+SAR
"END"COMFOUSER2;
  "EOP"

```



SECTION: 2,3

(MAY 1974)

PAGE 1

AUTHOR : H. FIOLET

INSTITUTE: MATHEMATICAL CENTRE,

RECEIVED: 731105.

**BRIEF DESCRIPTION:**

JFRAC CALCULATES A TERMINATING CONTINUED FRACTION.

**KEYWORDS:**

CONTINUED FRACTION,  
TERMINATING CONTINUED FRACTION.

**CALLING SEQUENCE:**

THE HEADING OF THE PROCEDURE READS:  
"REAL" "PROCEDURE" JFRAC(N,A,B);  
"VALUE" N; "INTEGER" N; "ARRAY" A,B;

JFRAC DELIVERS THE VALUE OF THE TERMINATING CONTINUED FRACTION:  
 $B[0] + A[1] / (B[1] + A[2] / (B[2] + A[3] / (B[3] + \dots + A[N] / B[N]))) \dots$

THE MEANING OF THE FORMAL PARAMETERS IS:

N: <ARITHMETIC EXPRESSION>;  
THE UPPER INDEX OF THE ARRAYS A AND B;

A,B: <ARRAY IDENTIFIER>;  
"ARRAY" A[1:N];  
"ARRAY" B[0:N];  
THE ELEMENTS OF THE CONTINUED FRACTION:  
 $B[0] + A[1] / (B[1] + A[2] / (B[2] + A[3] / (B[3] + \dots + A[N] / B[N]))) \dots$

PROCEDURES USED: NONE.

RUNNING TIME: PROPORTIONAL TO N.

LANGUAGE: ALGOL 60.



EXAMPLE OF USE:

```
"BEGIN"
"REAL" "PROCEDURE" JFRAC(N,A,B);"CODE" 35083;
"REAL" "ARRAY" P[1:10],Q[0:10];
"INTEGER" I;
"FOR" I:=1 "STEP" 1 "UNTIL" 10 "DO"
"BEGIN" P[I]:=1;Q[I]:=2 "END";
Q[0]:=1;
"FOR" I:=7 "STEP" 1 "UNTIL" 10 "DO"
OUTPUT(61,("N/"),JFRAC(I,P,Q))
"END"
```

DELIVERS:

```
+1.4142156862745"+000
+1.4142131979695"+000
+1.4142136248949"+000
+1.4142135516461"+000
```

SOURCE TEXT:

```
"CODE" 35083;
"REAL" "PROCEDURE" JFRAC(N,A,B);
"VALUE" N;"INTEGER" N;"ARRAY" A,B;
"BEGIN" "REAL" D;"INTEGER" I;
      D:=0;
      "FOR" I:=N "STEP" 1 "UNTIL" 1 "DO" D:=A[I]/(B[I]+D);
      JFRAC:=D+B[0]
"END" JFRAC;
"EOP"
```



SECTION: 2,4

(MAY 1974)

PAGE 1

AUTHOR: C.G. VAN DER LAAN,

INSTITUTE: MATHEMATICAL CENTRE,

RECEIVED: 730618,

**BRIEF DESCRIPTION:**

NEWGRN TRANSFORMS A POLYNOMIAL FROM THE NEWTON FORM INTO THE GRUNERT FORM.

**KEYWORDS:**

TRANSFORMATION,  
POLYNOMIAL REPRESENTATION,  
NEWTON,  
GRUNERT.

**CALLING SEQUENCE:**

THE HEADING OF THE PROCEDURE READS:  
"PROCEDURE" NEWGRN(N,X,C);  
"VALUE"N,"INTEGER"N,"ARRAY"X,C;

THE MEANING OF THE FORMAL PARAMETERS IS:  
N: <ARITHMETIC EXPRESSION>; THE DEGREE OF THE POLYNOMIAL;  
X: <ARRAY IDENTIFIER>;  
"ARRAY"X[0:N];  
THE INTERPOLATING POINTS; (SEE METHOD AND PERFORMANCE);  
C: <ARRAY IDENTIFIER>;  
"ARRAY"C[0:N];  
ENTRY: THE COEFFICIENTS OF THE NEWTON POLYNOMIAL FORM;  
EXIT: THE COEFFICIENTS OF THE GRUNERT POLYNOMIAL FORM.

PROCEDURES USED: ELMVEC = CP34020;

RUNNING TIME: THE NUMBER OF MULTIPLICATIONS IS  $N(N+1)$ .

LANGUAGE: ALGOL 60.



METHOD AND PERFORMANCE:

IN ARRAY X,C[0:N] ONE MUST GIVE THE VALUES  
 $X[I], I=0, \dots, N-1$ , AND  $C[I], I=0, \dots, N$   
 OF THE POLYNOMIAL OF (GIVEN) DEGREE N :  $P(Y) = C[0] + C[1]*(Y - X[0]) +$   
 $\dots + C[N]*(Y - X[0]) * \dots * (Y - X[N-1])$ .  
 THE COEFFICIENTS ,B[I], I=0, ..., N OF THE POLYNOMIAL IN THE  
 REPRESENTATION  $P(Y) = B[0] + B[1]*Y + \dots + B[N]*Y**N$ ,  
 ARE DELIVERED IN ARRAY C[0:N].

EXAMPLE OF USE:

```
"BEGIN" "ARRAY" X,F[0:2];
  "PROCEDURE" NEWGRN(N,X,C); "CODE" 31050;
  X[0]:=0;X[1]:=5;X[2]:=1;
  F[0]:=1;F[1]:=-2;F[2]:=2;
  NEWGRN(2,X,F);
  OUTPUT(61,"( "/,"( "THE GRUNERT COEFF. ARE" )",
    /,3(N)" )",F[0],F[1],F[2]);
"END" TSTNEWGRN;
```

THE GRUNERT COEFF. ARE

+1.0000000000000000"+000 -3.0000000000000000"+000 +2.0000000000000000"+000 .

SOURCE TEXT(S):

```
"CODE" 31050;
"PROCEDURE" NEWGRN(N,X,C);
"VALUE" N; "INTEGER" N; "ARRAY" X,C;
"BEGIN" "INTEGER" J,K,KM1; "REAL" XKM1,XXJ,XXJM1;
"ARRAY" XX[0:N];
"PROCEDURE" ELMVEC(L,U,SHIFT,A,B,X);
"CODE" 34020;
XX[0]:= 1; KM1:= 0;
"FOR" K:= 1 "STEP" 1 "UNTIL" N "DO"
"BEGIN" XX[K]:= 1; XKM1:= X[KM1];
  XXJ:=XX[KM1];
  "FOR" J:= KM1 "STEP" -1 "UNTIL" 1 "DO"
  "BEGIN" XXJM1:=XX[J-1];
    XX[J]:=XXJM1 - XXJ* XKM1;
    XXJ:=XXJM1;
  "END";
  XX[0]:= -XX[0] * XKM1;
  ELMVEC(0,KM1,0,C,XX,C[K]);
  KM1:=K;
"END"
"END" NEWGRN;
"EOP"
```



SECTION: 2.4.1

(DECEMBER 1975)

PAGE 1

AUTHOR: C.G. VAN DER LAAN.

INSTITUTE: MATHEMATICAL CENTRE.

RECEIVED: 730618.

BRIEF DESCRIPTION:

NEWGRN TRANSFORMS A POLYNOMIAL FROM THE NEWTON FORM INTO THE GRUNERT FORM.

KEYWORDS:

TRANSFORMATION,  
POLYNOMIAL REPRESENTATION,  
NEWTON,  
GRUNERT.

CALLING SEQUENCE:

THE HEADING OF THE PROCEDURE READS:

"PROCEDURE" NEWGRN(N,X,C);  
"VALUE" N; "INTEGER" N; "ARRAY" X,C;

THE MEANING OF THE FORMAL PARAMETERS IS:

N: <ARITHMETIC EXPRESSION>;  
THE DEGREE OF THE POLYNOMIAL;  
X: <ARRAY IDENTIFIER>;  
"ARRAY" X[0:N];  
THE INTERPOLATING POINTS: (SEE METHOD AND PERFORMANCE);  
C: <ARRAY IDENTIFIER>;  
"ARRAY" C[0:N];  
ENTRY: THE COEFFICIENTS OF THE NEWTON POLYNOMIAL FORM;  
EXIT: THE COEFFICIENTS OF THE GRUNERT POLYNOMIAL FORM.

PROCEDURES USED: ELMVEC = CP34020;

RUNNING TIME: THE NUMBER OF MULTIPLICATIONS IS  $N(N+1)$ .

LANGUAGE: ALGOL 60.



SECTION: 2.4.1

(DECEMBER 1975)

PAGE 2

## METHOD AND PERFORMANCE:

IN ARRAY X,C[0:N] ONE MUST GIVE THE VALUES X[I],I=0,...N-1, AND C[I],I=0,...N OF THE POLYNOMIAL OF (GIVEN) DEGREE N :  $P(Y)=C[0]+C[1]*(Y-X[0])+...+C[N]*(Y-X[0])*...*(Y-X[N-1])$ . THE COEFFICIENTS ,B[I],I=0,...N OF THE POLYNOMIAL IN THE REPRESENTATION  $P(Y)=B[0]+B[1]*Y+...+B[N]*Y**N$ , ARE DELIVERED IN ARRAY C[0:N].

## EXAMPLE OF USE:

```
"BEGIN" "ARRAY" X,F[0:2];
  "PROCEDURE" NEWGRN(N,X,C); "CODE" 31050;
  X[0]:=0;X[1]:=0.5;X[2]:=1;
  F[0]:=1;F[1]:=-2;F[2]:=2;
  NEWGRN(2,X,F);
  OUTPUT(61,"( "/, "(THE GRUNERT COEFF. ARE)",
    /,3(N)"",F[0],F[1],F[2]);
"END" TSTNEWGRN;
```

THE GRUNERT COEFF. ARE

```
+1.0000000000000000"+000 -3.0000000000000000"+000 +2.0000000000000000"+000 .
```

## SOURCE TEXT(S):

```
"CODE" 31050;
"PROCEDURE" NEWGRN(N,X,C);
"VALUE" N; "INTEGER" N; "ARRAY" X,C;
"BEGIN" "INTEGER" J,K,KM1; "REAL" XKM1,XXJ,XXJM1;
"ARRAY" XX[0:N];
"PROCEDURE" ELMVEC(L,U,SHIFT,A,B,X);
"CODE" 34020;
XX[0]:= 1; KM1:= 0;
"FOR" K:= 1 "STEP" 1 "UNTIL" N "DO"
"BEGIN" XX[K]:= 1; XKM1:= X[KM1];
  XXJ:=XX[KM1];
  "FOR" J:= KM1 "STEP" -1 "UNTIL" 1 "DO"
  "BEGIN" XXJM1:=XX[J-1];
    XX[J]:=XXJM1 - XXJ* XKM1;
    XXJ:=XXJM1;
  "END";
  XX[0]:= -XX[0] * XKM1;
  ELMVEC(0,KM1,0,C,XX,C[K]);
  KM1:=K;
"END"
"END" NEWGRN;
"EUP"
```



SECTION : 2,4,3

(OCTOBER 1974)

PAGE 1

AUTHOR : C.G. VAN DER LAAN.

INSTITUTE : RIJKSUNIVERSITEIT GRONINGEN.

RECEIVED : 740131.

BRIEF DESCRIPTION :

INTCHS COMPUTES THE INDEFINITE INTEGRAL OF A GIVEN CHEBYSHEV SERIES.

KEYWORDS :

INDEFINITE INTEGRATION,  
CHEBYSHEV SERIES.

CALLING SEQUENCE :

THE HEADING OF THE PROCEDURE READS :  
"PROCEDURE"INTCHS(N,A,B);  
"VALUE"N;"INTEGER"N;"ARRAY"A,B;

THE MEANING OF THE FORMAL PARAMETERS IS :

N : <ARITHMETIC EXPRESSION>;

ENTRY:

THE DEGREE OF THE POLYNOMIAL REPRESENTED BY THE CHEBYSHEV SERIES;

A,B: <ARRAY IDENTIFIER>;

"ARRAY" A[0:N],B[1:N+1];

ENTRY:

THE COEFFICIENTS OF THE CHEBYSHEV SERIES,  $A[0] + A[1] * T_1(X) + \dots + A[N] * T_N(X)$ , SHOULD BE GIVEN IN ARRAY A.

EXIT:

THE COEFFICIENTS OF THE INTEGRAL CHEBYSHEV SERIES,  $B[1] * T_1(X) + \dots + B[N+1] * T_{N+1}(X)$ , ARE DELIVERED IN ARRAY B. ( $T_1(X), \dots, T_{N+1}(X)$  DENOTE CHEBYSHEV POLYNOMIALS OF THE FIRST KIND, OF DEGREE 1, ..., N+1, RESPECTIVELY).



METHOD AND PERFORMANCE :

FOR A DESCRIPTION OF THE ALGORITHM SEE AMONG OTHERS :  
 CLENSHAW, 1962, P. 11, OR FOX AND PARKER, 1968, P. 59.

REFERENCES :

BROUCKE, R. (1973):  
 TEN SUBROUTINES FOR THE MANIPULATION OF CHEBYSHEV SERIES.  
 ALGORITHM 446. (FORTRAN).  
 COMM. ACM, VOL. 16, 1, P. 254-256.

CLENSHAW, C. W. (1962):  
 CHEBYSHEV SERIES FOR MATHEMATICAL FUNCTIONS.  
 MATH. TAB. NAT. PHYS. LAB. 5, LONDON.  
 H. M. STATIONARY OFFICE.

FOX, L. B. I. B. PARKER (1968):  
 CHEBYSHEV POLYNOMIALS IN NUMERICAL ANALYSIS.  
 OXFORD UNIVERSITY PRESS.

EXAMPLE OF USE :

AS A FORMAL TEST OF THE PROCEDURE INTCHS THE CHEBYSHEV SERIES :  
 $1 + 1/2 * T_1(X) + 1/5 * T_2(X) + 1/10 * T_3(X)$  IS TRANSFORMED INTO ITS INTEGRAL.

```
"BEGIN" "ARRAY" A[0:3], B[1:4];
  "PROCEDURE" INTCHS(N, A, B); "CODE" 31248;
  A[0] := 1; A[1] := .5; A[2] := .2; A[3] := "-1";
  INTCHS(3, A, B);
  OUTPUT(61, "(" / , 4(BZ, 4D) ")", B[1], B[2], B[3], B[4]);
"END"
```

.9000 .1000 .0333 .0125



## SOURCE TEXT(S):

```

"CODE"31248;
"PROCEDURE"INTCHS(N,A,B);
"VALUE"N,"INTEGER"N,"ARRAY"A,B;
"COMMENT"
    INTCHS DELIVERS THE COEFFICIENTS B[I], I=1,...,N+1, OF THE INTEGRAL
    CHEBYSHEV SERIES B[1]*T1(X)+...+B[N]*TN(X)+B[N+1]*TN+1(X).
    THESE COEFFICIENTS ARE OBTAINED BY MEANS OF INDEFINITE INTEGRATION
    OF THE CHEBYSHEV SERIES A[0]+A[1]*T1(X)+...+A[N]*TN(X),
    T1(X),...TN+1(X) DENOTE CHEBYSHEV POLYNOMIALS OF THE FIRST KIND,
    OF DEGREE 1,...,N+1, RESPECTIVELY;
"IF"N=0"THEN"B[1]:=A[0]
"ELSE""IF"N=1"THEN""BEGIN"B[2]:=A[1]/4;B[1]:=A[0]"END"
    "ELSE""BEGIN""INTEGER"I,"REAL"H,L,DUM;
        H:=A[N];DUM:=A[N-1];B[N+1]:=H/((N+1)*2);B[N]:=DUM/(N*2);
        "FOR"I:=N-1"STEP"-1"UNTIL"2"DO"
            "BEGIN"L:=A[I-1];B[I]:=(L-H)/(2*I);H:=DUM;DUM:=L
            "END";B[1]:=A[0]-H/2
    "END"INTCHS;
"EOP"

```